



Grafik 1: Rajah kasar dengan skala.



Grafik 2: Rajah berwarna (dipudarkan) dengan skala.




Kod lakaran (bulatan pada bahagian dalam anak bulan):

*Imej di atas yang terhasil daripada kod di bawah ini telah dikecilkan skalanya kepada 0.5cm per unit.

```
\startMPcode
draw image (
  picture Jalur; Jalur := image (
    draw fullsquare xysized (14,8);
    path p; p := fullsquare xscaled 14 shifted (0,3.5);
    for i= 0 downto - 7: draw p shifted (14 ,i); endfor;
    for i=-8 downto -13: draw p shifted (3.5,i) xscaled 2; endfor;
  ) scaled 1cm; draw Jalur withpen pencircle scaled .25pt;

  path Bulan; Bulan := fullcircle scaled 6cm;
  draw Bulan shifted (-1.25cm,0);
  draw Bulan scaled .889 shifted (-.583cm,0); fill fullsquare xyscaled
    (1.7cm,5cm) shifted (1.34cm,0) withcolor white; %to hide the overlapping lines

  Ra=1cm; Rb=2.5cm; d=360; n=28; % Bintang %
  for i=0 step 2 until 26:
    draw ((Ra*cosd(d*i/n),Ra*sind(d*i/n)) --
      (Rb*cosd(d*(i+1)/n),Rb*sind(d*(i+1)/n)) --
      (Ra*cosd(d*(i+2)/n),Ra*sind(d*(i+2)/n))) shifted (1.75cm,0);
  endfor;
  ) scaled .75;
\stopMPcode
```



Kod grafik berwarna (bulatan pada bahagian dalam anak bulan):

*Imej di atas yang terhasil daripada kod di bawah ini telah dikecilkan skalanya kepada 0.5cm per unit.

```
\definecolor[biru] [c=1,m=1.0,y=0,k=0]
\definecolor[merah] [c=0,m=.95,y=1,k=0]
\definecolor[kuning] [c=0,m=.09,y=1,k=0]

\startMPcode
draw image ( picture Jalur; Jalur := image (
  fill fullsquare xysized (14,8) withcolor \MPcolor{biru};
  path p; p:=fullsquare xscaled 14 shifted (0,3.5); forsuffices i=0,2,4,6:
  fill p shifted (14,-i ) withcolor \MPcolor{merah};
  fill p shifted (14,-i-1) withcolor white; endfor; forsuffices i=8,10,12:
  fill p shifted (3.5,-i ) xscaled 2 withcolor \MPcolor{merah};
  fill p shifted (3.5,-i-1) xscaled 2 withcolor white; endfor; )
  scaled 1cm; draw Jalur;

  path Bulan; Bulan := fullcircle scaled 6cm;
  fill Bulan shifted (-1.25cm,0) withcolor \MPcolor{kuning};
  fill Bulan scaled .889 shifted (-.583cm,0) withcolor \MPcolor{biru};

  Ra=1cm; Rb=2.5cm; d=360; n=28; path q; q:=fullcircle scaled 2cm; % Bintang %
  for i=0 step 2 until 26: fill ((Ra*cosd(d*i/n),Ra*sind(d*i/n)) --
    (Rb*cosd(d*(i+1)/n),Rb*sind(d*(i+1)/n))--(Ra*cosd(d*(i+2)/n),Ra*sind(d*(i+2)/n))
    -- origin -- cycle) shifted (1.75cm,0) withcolor \MPcolor{kuning};
  endfor; fill q shifted (1.75cm,0) withcolor \MPcolor{kuning}; %to hide image noise
) scaled .75; setbounds currentpicture to boundingbox currentpicture enlarged 3mm;
addbackground withcolor "lightgray";
\stopMPcode
```